Comparing various numerical evaluations for trigonometric integrals

```
> restart; interface(version);Digits:=14;
  infolevel[`evalf/int`]:=2:
  interface(rtablesize=20):
```

Standard Worksheet Interface, Maple 12.0, Windows XP, April 10 2008 Build ID 347164

$$Digits := 14 \quad (1)$$

Compare numerical results for the following integral and test values

```
> Int(x^k*exp(w*x*I),x = -1 .. 1);

  N:=5;
  wTst:=1/1000;
  L:='L':
```

$$\int_{-1}^{1} x^k e^{I w x} \, dx$$

$$N := 5$$

$$wTst := \frac{1}{1000} \quad (2)$$

Plotting suggests there should no problem (the imaginary part is re-scaled). For that test value the real part of the integral should be zero (since applied to an odd function).

```
> Int(x^k*exp(w*x*I),x = -1 .. 1); op(1,%): subs(k=N, w=wTst,%): evalc(%);
  plot( [Re(%),Im(%)/wTst], x=-1..1, color=[blue,red]);
```

$$\int_{-1}^{1} x^k e^{I w x} \, dx$$

$$x^5 \cos\left(\frac{1}{1000} x\right) + I x^5 \sin\left(\frac{1}{1000} x\right)$$



Except for the first two ways (and the very last one) the integral with its exact parameters is used.

1. Using lower case int with numerical bounds

That results in calling the NAG library for the real and imaginary part

```
> int(x^5*exp(wTst*x*I),x = -1.0 .. 1.0);
  L[1]:=%:
evalf/int/control: integrating on -1.0 .. 1.0 the integrand
```

$$x^5 e^{\frac{1}{1000} I x}$$

```
evalf/int/control: attempt to split integrand into Re and Im parts
evalf/int/control: integrating on -1.0 .. 1.0 the integrand
```

$$x^5 \cos\left(\frac{1}{1000} x\right)$$

```
Control: Entering NAGInt
Control: trying d01ajc (nag_1d_quad_gen)
d01ajc: trying evalhf callbacks
Control: d01ajc failed
evalf/int/control: NAG failed result = result
evalf/int/CreateProc: Trying easyproc
From ccquad, result = 0. integrand evals = 19 error = 0.
tolerance = .50000000000000e-16
evalf/int/control: integrating on -1.0 .. 1.0 the integrand
```

$$x^5 \sin\left(\frac{1}{1000} x\right)$$

```
Control: Entering NAGInt
Control: trying d01ajc (nag_1d_quad_gen)
d01ajc: trying evalhf callbacks
d01ajc: result=.285714248677250244e-3
d01ajc: abserr=.158603268672480492e-17; num_subint=1; fun_count=21
Control: result=.285714248677250244e-3
```

$$0. + 0.000285714248677250 I \quad (1.1)$$

2. Using lower case int, integrand made a numerical function and integer bounds

```
> int(x^5*exp(1.0*wTst*x*I),x = -1 .. 1);
  L[2]:=%:
```

$$60.659969670003 - 0.060659989890001 I \quad (2.1)$$

3. Evaluating symbolically and evaluate to a numerical number

```
> Int(x^5*exp(w*x*I),x = -1 .. 1); value(%);
  subs(w=wTst,%);
  evalf(%);
  L[3]:=%:
```

$$\int_{-1}^{1} x^5 e^{I w x}\, dx$$

$$-\frac{1}{w^6}\left(\left(120 + 120 I w - 60 w^2 - 20 I w^3 + 5 w^4 + I w^5 - 120 e^{2 I w} + 120 I e^{2 I w} w + 60 e^{2 I w} w^2 - 20 I e^{2 I w} w^3 - 5 e^{2 I w} w^4 + I e^{2 I w} w^5\right) e^{-I w}\right)$$

$$-1000000000000000000\left(\frac{23999988000001}{200000000000}+\frac{119999980000001}{1000000000000000}I\right.$$
$$\left.-\frac{23999988000001}{200000000000}e^{\frac{1}{500}I}+\frac{119999980000001}{1000000000000000}I\,e^{\frac{1}{500}I}\right)e^{-\frac{1}{1000}I}$$
$$39.999976666669-10000.044999992\,I \tag{3.1}$$

4. Evaluating symbolically, collect exponentials, and evaluate to a numerical number

```
> Int(x^5*exp(w*x*I),x = -1 .. 1); value(%):
  expand(%): combine(%,exp): normal(%): collect(%,exp);
  subs(w=wTst,%);
  evalf(%);
  L[4]:=%:
```

$$\int_{-1}^{1}x^5 e^{Iwx}\,dx$$
$$-\frac{\left(120+120\,I\,w-60\,w^2-20\,I\,w^3+5\,w^4+I\,w^5\right)e^{-Iw}}{w^6}-\frac{e^{Iw}\left(-120+120\,I\,w+60\,w^2-20\,I\,w^3-5\,w^4+I\,w^5\right)}{w^6}$$
$$(-1199999400000005000000-119999980000001000\,I)\,e^{-\frac{1}{1000}I}+(1199999400000005000000-119999980000001000\,I)\,e^{\frac{1}{1000}I}$$
$$0.-7999.9992000000\,I \tag{4.1}$$

5. Evaluating symbolically, simplify before inserting w and evaluate to a numerical number

```
> Int(x^5*exp(w*x*I),x = -1 .. 1); value(%); simplify(%);
  subs(w=wTst,%);
  evalf(%);
  L[5]:=%:
```

$$\int_{-1}^{1}x^5 e^{Iwx}\,dx$$
$$-\frac{1}{w^6}\Big(\big(120+120\,I\,w-60\,w^2-20\,I\,w^3+5\,w^4+I\,w^5-120\,e^{2Iw}+120\,I\,e^{2Iw}\,w+60\,e^{2Iw}\,w^2-20\,I\,e^{2Iw}\,w^3-5\,e^{2Iw}\,w^4+I\,e^{2Iw}\,w^5\big)\,e^{-Iw}\Big)$$
$$-\frac{1}{w^6}\big(120\,e^{-Iw}+120\,I\,e^{-Iw}\,w-60\,e^{-Iw}\,w^2-20\,I\,e^{-Iw}\,w^3+5\,e^{-Iw}\,w^4+I\,e^{-Iw}\,w^5-120\,e^{Iw}+120\,I\,e^{Iw}\,w+60\,e^{Iw}\,w^2-20\,I\,e^{Iw}\,w^3-5\,e^{Iw}\,w^4+I\,e^{Iw}\,w^5\big)$$
$$-1199999400000005000000\,e^{-\frac{1}{1000}I}-119999980000001000\,I\,e^{-\frac{1}{1000}I}$$

$$+ 11999994000000500000 e^{\frac{1}{1000} I} - 119999980000001000 I e^{\frac{1}{1000} I}$$

$$0. + 0. I \qquad (5.1)$$

6. Evaluating symbolically, simplify after inserting w and evaluate to a numerical number

```
> Int(x^5*exp(w*x*I),x = -1 .. 1); value(%);
  subs(w=wTst,%);
  simplify(%);
  evalf(%);
  L[6]:=%:
```

$$\int_{-1}^{1} x^5 e^{I w x} \, dx$$

$$-\frac{1}{w^6}\left(\left(120 + 120 I w - 60 w^2 - 20 I w^3 + 5 w^4 + I w^5 - 120 e^{2 I w} + 120 I e^{2 I w} w + 60 e^{2 I w} w^2 - 20 I e^{2 I w} w^3 - 5 e^{2 I w} w^4 + I e^{2 I w} w^5\right) e^{-I w}\right)$$

$$-1000000000000000000 \left(\frac{23999988000001}{200000000000} + \frac{119999980000001}{1000000000000000} I - \frac{23999988000001}{200000000000} e^{\frac{1}{500} I} + \frac{119999980000001}{1000000000000000} I e^{\frac{1}{500} I}\right) e^{-\frac{1}{1000} I}$$

$$2000 I \left(119999940000005000 \sin\left(\frac{1}{1000}\right) - 119999980000001 \cos\left(\frac{1}{1000}\right)\right)$$

$$0. I \qquad (6.1)$$

7. Symbolic evaluation and numerical through high precision

```
> Int(x^5*exp(w*x*I),x = -1 .. 1); value(%);
  subs(w=wTst,%);
  simplify(%);
  evalf[3*Digits](%): evalf(%);
  L[7]:=%:
```

$$\int_{-1}^{1} x^5 e^{I w x} \, dx$$

$$-\frac{1}{w^6}\left(\left(120 + 120 I w - 60 w^2 - 20 I w^3 + 5 w^4 + I w^5 - 120 e^{2 I w} + 120 I e^{2 I w} w + 60 e^{2 I w} w^2 - 20 I e^{2 I w} w^3 - 5 e^{2 I w} w^4 + I e^{2 I w} w^5\right) e^{-I w}\right)$$

$$-1000000000000000000 \left(\frac{23999988000001}{200000000000} + \frac{119999980000001}{1000000000000000} I - \frac{23999988000001}{200000000000} e^{\frac{1}{500} I} + \frac{119999980000001}{1000000000000000} I e^{\frac{1}{500} I}\right) e^{-\frac{1}{1000} I}$$

$$2000\,I\left(119999940000005000\sin\left(\frac{1}{1000}\right)-119999980000001\cos\left(\frac{1}{1000}\right)\right)$$

$$0.00028571424867725\,I \tag{7.1}$$

8. Symbolically, but decomposing into real and imaginary part

```
> Int(x^5*exp(w*x*I),x = -1 .. 1); map(evalc,%); expand(%); value(%);
  subs(w=wTst,%);
  evalf(%);
  L[8]:=%:
```

$$\int_{-1}^{1} x^5 e^{I w x}\,dx$$

$$\int_{-1}^{1} \left(x^5\cos(w x)+I x^5\sin(w x)\right)dx$$

$$\int_{-1}^{1} x^5\cos(w x)\,dx+I\left(\int_{-1}^{1} x^5\sin(w x)\,dx\right)$$

$$\frac{1}{w^6}\left(2\,I\left(-w^5\cos(w)+5\,w^4\sin(w)+20\,w^3\cos(w)-60\,w^2\sin(w)+120\sin(w)-120\cos(w)\,w\right)\right)$$

$$20000000000000000000\,I\left(-\frac{119999980000001}{1000000000000000}\cos\left(\frac{1}{1000}\right)+\frac{23999988000001}{200000000000}\sin\left(\frac{1}{1000}\right)\right)$$

$$0.\,I \tag{8.1}$$

A quite different way is to convert the monomial x^k into the Legendre base polynomials and use Bessel functions (due to a result of Bakhvalov & Vasileva),
we will re-discover some of the results already seen:

```
> 'Int( LegendreP(k,x) * exp(w*x*I),x = -1 .. 1)= BV(k,w)';
  BV:= (k,w) -> I^k*sqrt(2*Pi/w)*BesselJ(k+1/2,w);
```

$$\int_{-1}^{1} \mathrm{LegendreP}(k,x)\,e^{I w x}\,dx=\mathrm{BV}(k,w)$$

$$\mathrm{BV}:=(k,w)\to I^k\sqrt{\frac{2\pi}{w}}\,\mathrm{BesselJ}\left(k+\frac{1}{2},w\right) \tag{3}$$

```
> OrthogonalSeries:-ChangeBasis(x^N,LegendreP(n,x)): OrthogonalSeries:-
  ConvertToSum(%):
  x^N=convert(%,polynom);
  Int(x^5*exp(w*x*I),x = -1 .. 1) = '3/7*BV(1,w)+4/9*BV(3,w)+8/63*BV(5,w)';
```

$$x^5=\frac{3}{7}x+\frac{4}{9}\mathrm{LegendreP}(3,x)+\frac{8}{63}\mathrm{LegendreP}(5,x)$$

$$\int_{-1}^{1} x^5 e^{I w x}\, dx = \frac{3}{7} BV(1, w) + \frac{4}{9} BV(3, w) + \frac{8}{63} BV(5, w) \quad (4)$$

9. Using Bessel function and evaluate numerical

```
> '3/7*BV(1,w)+4/9*BV(3,w)+8/63*BV(5,w)';
  subs(w=wTst,%);
  evalf(%);
  L[9]:=%:
```

$$\frac{3}{7} BV(1, w) + \frac{4}{9} BV(3, w) + \frac{8}{63} BV(5, w)$$

$$\frac{6000000}{7} I \left(-\frac{1}{1000} \cos\left(\frac{1}{1000}\right) + \sin\left(\frac{1}{1000}\right)\right) + \frac{8000000000000}{9} I \left(\frac{14999999}{1000000000} \cos\left(\frac{1}{1000}\right) - \frac{7499997}{500000} \sin\left(\frac{1}{1000}\right)\right) + \frac{16000000000000000000}{63} I \left(-\frac{944999895000001}{1000000000000000} \cos\left(\frac{1}{1000}\right) + \frac{188999916000003}{200000000000} \sin\left(\frac{1}{1000}\right)\right)$$

$$2539.6828253968\, I \quad (9.1)$$

10. Using Bessel function, simplify and evaluate numerical

```
> '3/7*BV(1,w)+4/9*BV(3,w)+8/63*BV(5,w)';
  simplify(%);
  subs(w=wTst,%);
  evalf(%);
  L[10]:=%:
```

$$\frac{3}{7} BV(1, w) + \frac{4}{9} BV(3, w) + \frac{8}{63} BV(5, w)$$

$$\frac{1}{w^{11/2}} \left(2 I \sqrt{\frac{1}{w}} \left(-w^5 \cos(w) + 5 w^4 \sin(w) + 20 w^3 \cos(w) - 60 w^2 \sin(w) + 120 \sin(w) - 120 \cos(w)\, w\right)\right)$$

$$2000000000000000000\, I \left(-\frac{119999980000001}{1000000000000000} \cos\left(\frac{1}{1000}\right) + \frac{23999988000001}{200000000000} \sin\left(\frac{1}{1000}\right)\right)$$

$$0.\, I \quad (10.1)$$

11. Using Bessel function with numerical arguments and evaluate numerical

```
> '3/7*BV(1.0,w)+4/9*BV(3.0,w)+8/63*BV(5.0,w)';
  subs(w=wTst,%);
  evalf(%);
  L[11]:=%:
```

$$\frac{3}{7}\,\mathrm{BV}(1.0, w) + \frac{4}{9}\,\mathrm{BV}(3.0, w) + \frac{8}{63}\,\mathrm{BV}(5.0, w)$$

$$0.42857142857143\,I\sqrt{2}\,\sqrt{1000}\,\sqrt{\pi}\,\mathrm{BesselJ}\left(1.5000000000000, \frac{1}{1000}\right) - 0.44444444444444\,I\sqrt{2}\,\sqrt{1000}\,\sqrt{\pi}\,\mathrm{BesselJ}\left(3.5000000000000, \frac{1}{1000}\right) + 0.12698412698413\,I\sqrt{2}\,\sqrt{1000}\,\sqrt{\pi}\,\mathrm{BesselJ}\left(5.5000000000000, \frac{1}{1000}\right)$$

$$0.00028571424867725\,I \tag{11.1}$$

Cleaning the numerics (using fnormal and indicative zeros) for the 11 ways we have 6 different numerical results. The first one (using NAG) and the last one (Bessel with numerical inputs) are the only correct results.

```
> #convert(L,list): convert(%,Vector);
  convert(L,set): map(fnormal,%): simplify(%,zero): convert(%,list): sort(%):
  convert(%,Vector);
  cat(nops(%%),  ` different values`);
```

$$\begin{bmatrix} 0. \\ 39.999976666669 - 10000.044999992\,I \\ -7999.9992000000\,I \\ 60.659969670003 - 0.060659989890001\,I \\ 2539.6828253968\,I \\ 0.00028571424867725\,I \end{bmatrix}$$

$$\text{6 different values} \tag{5}$$